Generating Function for Legendre Polynomials

```
> Phi := (x,h) -> 1/sqrt(1-2*x*h+h^2);
```

$$\Phi := (x, h) \to \frac{1}{\sqrt{1-2\,x\,h+h^2}} \quad (1)$$

```
> z:=taylor(Phi(x,h),h=0,7);
```

$$z := 1 + x\,h + \left(-\frac{1}{2} + \frac{3}{2}x^2\right)h^2 + \left(-\frac{3}{2}x + \frac{5}{2}x^3\right)h^3 + \left(\frac{3}{8} - \frac{15}{4}x^2 + \frac{35}{8}x^4\right)h^4 + \left(\frac{15}{8}x - \frac{35}{4}x^3 + \frac{63}{8}x^5\right)h^5 + \left(-\frac{5}{16} + \frac{105}{16}x^2 - \frac{315}{16}x^4 + \frac{231}{16}x^6\right)h^6 + \mathrm{O}(h^7) \quad (2)$$

```
> with(orthopoly);
  P(6,x);
```

$$[G, H, L, P, T, U]$$

$$-\frac{5}{16} + \frac{231}{16}x^6 - \frac{315}{16}x^4 + \frac{105}{16}x^2 \quad (3)$$

Check of a few coefficients of h^l in Phi(x,h) being P_l(x)

```
> for j from 1 to 6 do
  P(j,x)-coeff(z,h^j);
  od;
```

$$0$$
$$0$$
$$0$$
$$0$$
$$0$$
$$0 \quad (4)$$

To show coefficients of h^l in Phi(x,h) is P_l(x) up to a constant multiple

```
> (1-x^2)*diff(Phi(x,h),x$2)-2*x*diff(Phi(x,h),x)+h*diff(h*Phi(x,
  h),h$2);
```

$$\frac{3\,(1-x^2)\,h^2}{(1-2\,x\,h+h^2)^{5/2}} - \frac{2\,x\,h}{(1-2\,x\,h+h^2)^{3/2}} + h\left(-\frac{-2\,x+2\,h}{(1-2\,x\,h+h^2)^{3/2}} + \frac{3}{4}\,\frac{h\,(-2\,x+2\,h)^2}{(1-2\,x\,h+h^2)^{5/2}} - \frac{h}{(1-2\,x\,h+h^2)^{3/2}}\right) \quad (5)$$

```
> expand(%);
```

$$\frac{3\,h^2}{(1-2\,x\,h+h^2)^{5/2}} - \frac{3\,h^2}{(1-2\,x\,h+h^2)^{3/2}} - \frac{6\,h^3\,x}{(1-2\,x\,h+h^2)^{5/2}} + \frac{3\,h^4}{(1-2\,x\,h+h^2)^{5/2}} \quad (6)$$

```
> simplify(%);
```

$$0 \quad (7)$$

Relation (1-2*x*h+h^2)*diff(Phi(x,h),h) = (x-h)*Phi(x,h)

```
> diff(Phi(x,h),h);
```

$$-\frac{-x+h}{\left(1-2\,x\,h+h^2\right)^{3/2}} \tag{8}$$

To show x*P_l'-P'_{l-1}=l*P_l

```
> x*diff(Phi(x,h),x)-diff(h*Phi(x,h),x)-h*diff(Phi(x,h),h);
```

$$\frac{x\,h}{\left(1-2\,x\,h+h^2\right)^{3/2}}-\frac{h^2}{\left(1-2\,x\,h+h^2\right)^{3/2}}+\frac{1}{2}\,\frac{h\,(-2\,x+2\,h)}{\left(1-2\,x\,h+h^2\right)^{3/2}} \tag{9}$$

```
> expand(%);
```

$$0 \tag{10}$$

To show P_l'-x*P'_{l-1}=l*P_{l-1}

```
> diff(Phi(x,h),x)-x*diff(h*Phi(x,h),x)-h*diff(h*Phi(x,h),h);
```

$$\frac{h}{\left(1-2\,x\,h+h^2\right)^{3/2}}-\frac{x\,h^2}{\left(1-2\,x\,h+h^2\right)^{3/2}}-h\left(\frac{1}{\sqrt{1-2\,x\,h+h^2}}-\frac{1}{2}\,\frac{h\,(-2\,x+2\,h)}{\left(1-2\,x\,h+h^2\right)^{3/2}}\right) \tag{11}$$

```
> simplify(expand(%));
```

$$0 \tag{12}$$

A few plots of Legendre polynomials

```
> plot(P(3,x),x=-1..1);
```




```
> plot(P(24,x),x=-1..1);
```